
File created:  15-Jun-90 11:50:26  {DSK}<usr>local>lde>lispcore>internal>library>CMLHELP.;2

changes to:  (IL:VARS IL:CMLHELPCOMS)

previous date:  20-Oct-86 11:42:05  {DSK}<usr>local>lde>lispcore>internal>library>CMLHELP.;1

Read Table:  XCL

Package:  LISP

Format:  XCCS

; Copyright (c) 1985, 1986, 1990 by Venue & Xerox Corporation.  All rights reserved.

(IL:RPAQQ IL:CMLHELPCOMS ((IL:INITVARS (IL:CMLPATH ’(IL:|{ERIS}<CommonLoops>PCL>| IL:{ERIS}<COMMONLISP>CODE> 
IL:{ERIS}<COMMONLISP>CLC>))

(IL:CMLWINDOW))
(IL:FNS IL:ADDHASHFILE IL:CMLSHOW IL:MAKECMLINDEX IL:CMLWINDOW IL:CMLHASHFILE)
(IL:INITVARS (IL:CMLDEFS)

(IL:CMLMANHASH))))

(IL:RPAQ? IL:CMLPATH ’(IL:|{ERIS}<CommonLoops>PCL>| IL:{ERIS}<COMMONLISP>CODE> IL:{ERIS}<COMMONLISP>CLC>))

(IL:RPAQ? IL:CMLWINDOW )

(IL:DEFINEQ

(IL:ADDHASHFILE
  (IL:LAMBDA (IL:NAME IL:VAL IL:HARRAY)

(IL:PUTHASHFILE IL:NAME (CONS IL:VAL (IL:GETHASHFILE IL:NAME IL:HARRAY))
IL:HARRAY)))

(IL:CMLSHOW
  (IL:LAMBDA (IL:NAME IL:TO IL:MANP) (IL:* IL:|lmm| " 9-May-86 17:13")

(LET
((IL:OUT (OR IL:TO (IL:|if| IL:MANP

IL:|then| (IL:SETQ IL:TO T)
IL:|else| (IL:OPENSTREAM ’IL:{SCRATCH} ’IL:BOTH ’IL:NEW))))

IL:WINDOW
(IL:FOUND 0)
IL:STR)
(IL:|for| IL:HF IL:|inside| (IL:CMLHASHFILE IL:MANP)

IL:|do|
(LET ((IL:LST (IL:GETHASHFILE (IL:U-CASE IL:NAME)

(CAR IL:HF))))
(IL:|for| IL:|occurence| IL:|inside| (REMOVE-DUPLICATES IL:LST :TEST ’IL:EQUAL)

IL:|do| (IL:RESETLST
(IL:RESETSAVE NIL (LIST ’IL:CLOSEF? (IL:SETQ IL:STR

(IL:|for| IL:FILE IL:|inside| (CDR IL:HF) IL:|bind| IL:FN
IL:|when| (IL:SETQ IL:FN

(IL:INFILEP (IL:PACKFILENAME
’IL:BODY
(CAR IL:|occurence|)
’IL:BODY IL:FILE)))

IL:|do| (RETURN (IL:OPENSTREAM IL:FN
’IL:INPUT
’IL:OLD))))))

(IL:|if| IL:STR
IL:|then| (IL:* IL:|we| IL:|found| IL:|the| IL:|file| IL:|on| IL:|this| IL:|dir|)

(LET ((IL:END (OR (IL:FILEPOS (IL:|if| IL:MANP
IL:|then| 

(IL:* IL:|end| IL:|of| IL:|definition|)
"
@ENDDEF"

IL:|else| 
(IL:* IL:|dunno| IL:|where| IL:|it| IL:|ends,| IL:|just| IL:|get| IL:|the|

IL:|next| IL:|one|)
"
(def")

IL:STR
(IL:PLUS 4 (CADR IL:|occurence|))
NIL NIL NIL (IL:UPPERCASEARRAY))

(IL:GETEOFPTR IL:STR))))
(IL:SELECTQ IL:OUT

(IL:TEDIT (IL:OPENTEXTSTREAM IL:STR (IL:|if| IL:WINDOW
IL:|then| (IL:CREATEW NIL 

"CML
definitions

")
IL:|else| (IL:SETQ IL:WINDOW

(IL:CMLWINDOW)))
(CADR IL:|occurence|)
IL:END
(AND (NOT IL:MANP)

’(IL:FONT (IL:TERMINAL 10)
IL:PARALOOKS
(IL:TABS (48))))))

((IL:ALLFILE T) 
(IL:SETQ IL:WINDOW (IL:|if| IL:WINDOW

IL:|then| (IL:CREATEW NIL "CML
definitions")

IL:|else| (IL:CMLWINDOW)))
(IL:WINDOWPROP IL:WINDOW ’IL:TITLE (IL:FULLNAME IL:STR))
(IL:TEDIT.SETSEL
(IL:SETQ IL:STR (IL:OPENTEXTSTREAM

IL:STR NIL NIL NIL
(AND (NOT IL:MANP)

’(IL:FONT (IL:TERMINAL 10)
IL:PARALOOKS
(IL:TABS (48))))))

(CADR IL:|occurence|)
(IL:DIFFERENCE IL:END (CADR IL:|occurence|)))
(IL:OPENTEXTSTREAM IL:STR IL:WINDOW NIL NIL

(AND (NOT IL:MANP)
’(IL:FONT (IL:TERMINAL 10)

IL:PARALOOKS
(IL:TABS (48))))))

(PROGN (IL:PRINTOUT IL:OUT "(from " (IL:FULLNAME IL:STR)
")" T)

(IL:COPYBYTES IL:STR IL:OUT (CADR IL:|occurence|)
IL:END)

(IL:TERPRI IL:OUT)))
(IL:|add| IL:FOUND 1)))))))

(IL:|if| (NOT IL:TO)
IL:|then| (IL:OPENTEXTSTREAM IL:OUT (LET ((IL:W (IL:CMLWINDOW)))

(IL:WINDOWPROP IL:W ’IL:TITLE "Common Lisp definition")
IL:W)

NIL NIL ’(IL:FONT (IL:TERMINAL 10)
IL:PARALOOKS
(IL:TABS (48))))

(LET ((IL:PW (IL:GETPROMPTWINDOW (IL:CMLWINDOW))))
(IL:CLEARW IL:PW)
(IL:SELECTQ IL:FOUND

(0 (IL:PRINTOUT IL:PW "No occurences of" IL:\, IL:NAME "."))
(1 (IL:PRINTOUT IL:PW IL:FOUND IL:\, "occurence of" IL:\, IL:NAME "."))
(IL:PRINTOUT IL:PW IL:FOUND IL:\, "occurences of" IL:\, IL:NAME "."))))

IL:FOUND)))

(IL:MAKECMLINDEX
  (IL:LAMBDA (IL:PATTERN IL:HF IL:MANP) (IL:* IL:|lmm| "28-Apr-86 11:44")

(IL:SETQ IL:HF (IL:OPENHASHFILE (OR IL:HF "CML.HASH")
’IL:NEW 40 3080))

(IL:|bind| IL:STR IL:NF (IL:READTABLE IL:_ (IL:|if| IL:MANP
IL:|then| (LET ((IL:RT (IL:COPYREADTABLE IL:CMLRDTBL)))

(IL:SETSYNTAX (IL:CHARCODE ",")
’IL:SEPR IL:RT)

(IL:SETSYNTAX (IL:CHARCODE IL:})
’IL:SEPR IL:RT)

(IL:SETSYNTAX (IL:CHARCODE IL:{)
’IL:SEPR IL:RT)

IL:RT)
IL:|else| IL:CMLRDTBL))

IL:|for| IL:FILE IL:|in| (IL:FILDIR (IL:PACKFILENAME ’IL:BODY IL:PATTERN ’IL:EXTENSION ’IL:* ’IL:VERSION ""
’IL:NAME
’IL:*))

IL:|do| (IL:RESETLST
(IL:RESETSAVE NIL (LIST ’IL:CLOSEF? (IL:SETQ IL:STR (IL:OPENSTREAM IL:FILE ’IL:INPUT

’IL:OLD))))
(IL:SETQ IL:NF (IL:NAMEFIELD IL:FILE T))
(IL:PRINTOUT T IL:FILE T)
(IL:|while| (IL:FILEPOS (IL:|if| IL:MANP

IL:|then| "
@Def"

IL:|else| "
(def")

IL:STR NIL NIL NIL NIL (IL:UPPERCASEARRAY))
IL:|do| (IL:READC IL:STR)

(LET ((IL:POS (IL:GETFILEPTR IL:STR))
(IL:DEFFER (PROGN (IL:BIN IL:STR)

(IL:READ IL:STR IL:READTABLE)))
(IL:DEFD (IL:READ IL:STR IL:READTABLE)))

(IL:PRINTOUT T ".")
(LET ((IL:DEFA (IL:|if| (IL:LISTP IL:DEFD)

IL:|then| (IL:|if| IL:MANP
IL:|then| (CADR IL:DEFD)

IL:|else| (CAR IL:DEFD))
IL:|else| IL:DEFD)))

(IL:|if| (SYMBOLP IL:DEFA)
IL:|then| (IL:ADDHASHFILE IL:DEFA (LIST IL:NF IL:POS)

IL:HF)))))))))

(IL:CMLWINDOW

  (IL:LAMBDA NIL
(OR IL:CMLWINDOW (IL:SETQ IL:CMLWINDOW (IL:CREATEW ’(0 0 500 300)

"CML definitions")))))

(IL:CMLHASHFILE
  (IL:LAMBDA (IL:MANP) (IL:* IL:|lmm| "28-Apr-86 11:59")

(IL:|if| IL:MANP
IL:|then| (OR IL:CMLMANHASH (IL:SETQ IL:CMLMANHASH (LIST (CONS (IL:OPENHASHFILE 

’IL:{ERIS}<COMMONLISP>MANUAL>CML.HASH
’IL:INPUT)

’IL:{ERIS}<COMMONLISP>MANUAL>))))
IL:|else| (OR IL:CMLDEFS (IL:SETQ IL:CMLDEFS (IL:|for| IL:|path| IL:|inside| IL:CMLPATH IL:|bind| IL:FILE

IL:|when| (IL:SETQ IL:FILE (IL:FINDFILE "CML.HASH" T
(IL:MKLIST IL:|path|)))

IL:|collect| (CONS (IL:OPENHASHFILE IL:FILE ’IL:INPUT)
IL:|path|)))))))

)

(IL:RPAQ? IL:CMLDEFS )

(IL:RPAQ? IL:CMLMANHASH )

(IL:PUTPROPS IL:CMLHELP IL:COPYRIGHT ("Venue & Xerox Corporation" 1985 1986 1990))

-- Listed on  9-Oct-2024 02:45:48 --

FUNCTION INDEX

IL:ADDHASHFILE ....1 IL:CMLHASHFILE ....3 IL:CMLSHOW ........1 IL:CMLWINDOW ......2 IL:MAKECMLINDEX ...2

VARIABLE INDEX

IL:CMLDEFS ........3 IL:CMLMANHASH .....3 IL:CMLPATH ........1 IL:CMLWINDOW ......1


